

# NAG Library Routine Document

## F02JQF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F02JQF solves the quadratic eigenvalue problem

$$(\lambda^2 A + \lambda B + C)x = 0,$$

where  $A$ ,  $B$  and  $C$  are complex  $n$  by  $n$  matrices.

The routine returns the  $2n$  eigenvalues,  $\lambda_j$ , for  $j = 1, 2, \dots, 2n$ , and can optionally return the corresponding right eigenvectors,  $x_j$  and/or left eigenvectors,  $y_j$  as well as estimates of the condition numbers of the computed eigenvalues and backward errors of the computed right and left eigenvectors. A left eigenvector satisfies the equation

$$y^H(\lambda^2 A + \lambda B + C) = 0,$$

where  $y^H$  is the complex conjugate transpose of  $y$ .

$\lambda$  is represented as the pair  $(\alpha, \beta)$ , such that  $\lambda = \alpha/\beta$ . Note that the computation of  $\alpha/\beta$  may overflow and indeed  $\beta$  may be zero.

### 2 Specification

```

SUBROUTINE F02JQF (SCAL, JOBVL, JOBVR, SENSE, TOL, N, A, LDA, B, LDB, C,      &
                  LDC, ALPHA, BETA, VL, LDVL, VR, LDVR, S, BEVL, BEVR,    &
                  IWARN, IFAIL)
INTEGER          SCAL, SENSE, N, LDA, LDB, LDC, LDVL, LDVR, IWARN,      &
                  IFAIL
REAL (KIND=nag_wp) TOL, S(*), BEVL(*), BEVR(*)
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), C(LDC,*), ALPHA(2*N),      &
                  BETA(2*N), VL(LDVL,*), VR(LDVR,*)
CHARACTER(1)     JOBVL, JOBVR

```

### 3 Description

The quadratic eigenvalue problem is solved by linearizing the problem and solving the resulting  $2n$  by  $2n$  generalized eigenvalue problem. The linearization is chosen to have favourable conditioning and backward stability properties. An initial preprocessing step is performed that reveals and deflates the zero and infinite eigenvalues contributed by singular leading and trailing matrices.

The algorithm is backward stable for problems that are not too heavily damped, that is  $\|B\| \leq \sqrt{\|A\| \cdot \|C\|}$ .

Further details on the algorithm are given in Hammarling *et al.* (2013).

### 4 References

Fan H -Y, Lin W.-W and Van Dooren P. (2004) Normwise scaling of second order polynomial matrices. *SIAM J. Matrix Anal. Appl.* **26**, 1 252–256

Gaubert S and Sharify M (2009) Tropical scaling of polynomial matrices *Lecture Notes in Control and Information Sciences Series* **389** 291–303 Springer–Verlag

Hammarling S, Munro C J and Tisseur F (2013) An algorithm for the complete solution of quadratic eigenvalue problems. *ACM Trans. Math. Software.* **39(3):18:1–18:119** (<http://eprints.ma.man.ac.uk/1815/>)

## 5 Parameters

- 1: SCAL – INTEGER *Input*
- On entry:* determines the form of scaling to be performed on  $A$ ,  $B$  and  $C$ .
- SCAL = 0  
No scaling.
- SCAL = 1 (the recommended value)  
Fan, Lin and Van Dooren scaling if  $\frac{\|B\|}{\sqrt{\|A\| \times \|C\|}} < 10$  and no scaling otherwise where  $\|Z\|$  is the Frobenius norm of  $Z$ .
- SCAL = 2  
Fan, Lin and Van Dooren scaling.
- SCAL = 3  
Tropical scaling with largest root.
- SCAL = 4  
Tropical scaling with smallest root.
- Constraint:* SCAL = 0, 1, 2, 3 or 4.
- 2: JOBVL – CHARACTER(1) *Input*
- On entry:* if JOBVL = 'N', do not compute left eigenvectors.  
If JOBVL = 'V', compute the left eigenvectors.  
If SENSE = 1, 2, 4, 5, 6 or 7, JOBVL must be set to 'V'.  
*Constraint:* JOBVL = 'N' or 'V'.
- 3: JOBVR – CHARACTER(1) *Input*
- On entry:* if JOBVR = 'N', do not compute right eigenvectors.  
If JOBVR = 'V', compute the right eigenvectors.  
If SENSE = 1, 3, 4, 5, 6 or 7, JOBVR must be set to 'V'.  
*Constraint:* JOBVR = 'N' or 'V'.
- 4: SENSE – INTEGER *Input*
- On entry:* determines whether, or not, condition numbers and backward errors are computed.
- SENSE = 0  
Do not compute condition numbers, or backward errors.
- SENSE = 1  
Just compute condition numbers for the eigenvalues.
- SENSE = 2  
Just compute backward errors for the left eigenpairs.
- SENSE = 3  
Just compute backward errors for the right eigenpairs.
- SENSE = 4  
Compute backward errors for the left and right eigenpairs.
- SENSE = 5  
Compute condition numbers for the eigenvalues and backward errors for the left eigenpairs.
- SENSE = 6  
Compute condition numbers for the eigenvalues and backward errors for the right eigenpairs.

SENSE = 7

Compute condition numbers for the eigenvalues and backward errors for the left and right eigenpairs.

*Constraint:* SENSE = 0, 1, 2, 3, 4, 5, 6 or 7.

- 5: TOL – REAL (KIND=nag\_wp) *Input*  
*On entry:* TOL is used as the tolerance for making decisions on rank in the deflation procedure. If TOL is zero on entry then  $n \times \max(\|A\|, \|B\|, \|C\|) \times \textit{machine precision}$  is used in place of TOL, where  $\|Z\|$  is the Frobenius norm of the (scaled) matrix  $Z$  and *machine precision* is as returned by routine X02AJF. If TOL is  $-1.0$  on entry then no deflation is attempted. The recommended value for TOL is zero.
- 6: N – INTEGER *Input*  
*On entry:* the order of the matrices  $A$ ,  $B$  and  $C$ .  
*Constraint:*  $N \geq 0$ .
- 7: A(LDA, \*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $N$ .  
*On entry:* the  $n$  by  $n$  matrix  $A$ .  
*On exit:*  $A$  is used as internal workspace, but if  $\text{JOBVL} = 'V'$  or  $\text{JOBVR} = 'V'$ , then  $A$  is restored on exit.
- 8: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F02JQF is called.  
*Constraint:*  $LDA \geq N$ .
- 9: B(LDB, \*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $B$  must be at least  $N$ .  
*On entry:* the  $n$  by  $n$  matrix  $B$ .  
*On exit:*  $B$  is used as internal workspace, but is restored on exit.
- 10: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array  $B$  as declared in the (sub)program from which F02JQF is called.  
*Constraint:*  $LDB \geq N$ .
- 11: C(LDC, \*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $C$  must be at least  $N$ .  
*On entry:* the  $n$  by  $n$  matrix  $C$ .  
*On exit:*  $C$  is used as internal workspace, but if  $\text{JOBVL} = 'V'$  or  $\text{JOBVR} = 'V'$ ,  $C$  is restored on exit.
- 12: LDC – INTEGER *Input*  
*On entry:* the first dimension of the array  $C$  as declared in the (sub)program from which F02JQF is called.  
*Constraint:*  $LDC \geq N$ .

- 13: ALPHA( $2 \times N$ ) – COMPLEX (KIND=nag\_wp) array Output  
*On exit:* ALPHA( $j$ ), for  $j = 1, 2, \dots, 2n$ , contains the first part of the the  $j$ th eigenvalue pair  $(\alpha_j, \beta_j)$  of the quadratic eigenvalue problem.
- 14: BETA( $2 \times N$ ) – COMPLEX (KIND=nag\_wp) array Output  
*On exit:* BETA( $j$ ), for  $j = 1, 2, \dots, 2n$ , contains the second part of the  $j$ th eigenvalue pair  $(\alpha_j, \beta_j)$  of the quadratic eigenvalue problem. Although BETA is declared complex, it is actually real and non-negative. Infinite eigenvalues have  $\beta_j$  set to zero.
- 15: VL(LDVL,\*) – COMPLEX (KIND=nag\_wp) array Output  
**Note:** the second dimension of the array VL must be at least  $2 \times N$  if JOBVL = 'V'.  
*On exit:* if JOBVL = 'V', the left eigenvectors  $y_j$  are stored one after another in the columns of VL, in the same order as the corresponding eigenvalues. Each eigenvector will be normalized with length unity and with the element of largest modulus real and positive.  
 If JOBVL = 'N', VL is not referenced.
- 16: LDVL – INTEGER Input  
*On entry:* the first dimension of the array VL as declared in the (sub)program from which F02JQF is called.  
*Constraint:* LDVL  $\geq$  N.
- 17: VR(LDVR,\*) – COMPLEX (KIND=nag\_wp) array Output  
**Note:** the second dimension of the array VR must be at least  $2 \times N$  if JOBVR = 'V'.  
*On exit:* if JOBVR = 'V', the right eigenvectors  $x_j$  are stored one after another in the columns of VR, in the same order as the corresponding eigenvalues. Each eigenvector will be normalized with length unity and with the element of largest modulus real and positive.  
 If JOBVR = 'N', VR is not referenced.
- 18: LDVR – INTEGER Input  
*On entry:* the first dimension of the array VR as declared in the (sub)program from which F02JQF is called.  
*Constraint:* LDVR  $\geq$  N.
- 19: S(\*) – REAL (KIND=nag\_wp) array Output  
**Note:** the dimension of the array S must be at least  $2 \times N$  if SENSE = 1, 5, 6 or 7.  
**Note:** also: computing the condition numbers of the eigenvalues requires that both the left and right eigenvectors be computed.  
*On exit:* if SENSE = 1, 5, 6 or 7, S( $j$ ) contains the condition number estimate for the  $j$ th eigenvalue (large condition numbers imply that the problem is near one with multiple eigenvalues). Infinite condition numbers are returned as the largest model real number (X02ALF).  
 If SENSE = 0, 2, 3 or 4, S is not referenced.
- 20: BEVL(\*) – REAL (KIND=nag\_wp) array Output  
**Note:** the dimension of the array BEVL must be at least  $2 \times N$  if SENSE = 2, 4, 5 or 7.  
*On exit:* if SENSE = 2, 4, 5 or 7, BEVL( $j$ ) contains the backward error estimate for the computed left eigenpair  $(\lambda_j, y_j)$ .  
 If SENSE = 0, 1, 3 or 6, BEVL is not referenced.

21: BEVR(\*) – REAL (KIND=nag\_wp) array Output

**Note:** the dimension of the array BEVR must be at least  $2 \times N$  if SENSE = 3, 4, 6 or 7.

*On exit:* if SENSE = 3, 4, 6 or 7, BEVR(*j*) contains the backward error estimate for the computed right eigenpair  $(\lambda_j, x_j)$ .

If SENSE = 0, 1, 2 or 5, BEVR is not referenced.

22: IWARN – INTEGER Output

*On exit:* IWARN will be positive if there are warnings, otherwise IWARN will be 0.

If IFAIL = 0 then:

if IWARN = 1 then one, or both, of the matrices *A* and *C* is zero. In this case no scaling is performed, even if SCAL > 0;

if IWARN = 2 then the matrices *A* and *C* are singular, or nearly singular, so the problem is potentially ill-posed;

if IWARN = 3 then both the conditions for IWARN = 1 and IWARN = 2 above, apply.

If IFAIL = 2, IWARN returns the value of INFO from F08XNF (ZGGES).

If IFAIL = 3, IWARN returns the value of INFO from F08WNF (ZGGEV).

23: IFAIL – INTEGER Input/Output

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The quadratic matrix polynomial is nonregular (singular).

IFAIL = 2

The *QZ* iteration failed in F08XNF (ZGGES).

IWARN returns the value of INFO returned by F08XNF (ZGGES). This failure is unlikely to happen, but if it does, please contact NAG.

IFAIL = 3

The *QZ* iteration failed in F08WNF (ZGGEV).

IWARN returns the value of INFO returned by F08WNF (ZGGEV). This failure is unlikely to happen, but if it does, please contact NAG.

IFAIL = -1

On entry, SCAL =  $\langle value \rangle$ .  
Constraint: SCAL = 0, 1, 2, 3 or 4.

IFAIL = -2

On entry, JOBVL =  $\langle value \rangle$ .  
Constraint: JOBVL = 'N' or 'V'.  
On entry, SENSE =  $\langle value \rangle$  and JOBVL =  $\langle value \rangle$ .  
Constraint: when JOBVL = 'N', SENSE = 0 or 3,  
when JOBVL = 'V', SENSE = 1, 2, 4, 5, 6 or 7.

IFAIL = -3

On entry, JOBVR =  $\langle value \rangle$ .  
Constraint: JOBVR = 'N' or 'V'.  
On entry, SENSE =  $\langle value \rangle$  and JOBVR =  $\langle value \rangle$ .  
Constraint: when JOBVR = 'N', SENSE = 0 or 2,  
when JOBVR = 'V', SENSE = 1, 3, 4, 5, 6 or 7.

IFAIL = -4

On entry, SENSE =  $\langle value \rangle$ .  
Constraint: SENSE = 0, 1, 2, 3, 4, 5, 6 or 7.

IFAIL = -6

On entry, N =  $\langle value \rangle$ .  
Constraint:  $N \geq 0$ .

IFAIL = -8

On entry, LDA =  $\langle value \rangle$  and N =  $\langle value \rangle$ .  
Constraint:  $LDA \geq N$ .

IFAIL = -10

On entry, LDB =  $\langle value \rangle$  and N =  $\langle value \rangle$ .  
Constraint:  $LDB \geq N$ .

IFAIL = -12

On entry, LDC =  $\langle value \rangle$  and N =  $\langle value \rangle$ .  
Constraint:  $LDC \geq N$ .

IFAIL = -16

On entry, LDVL =  $\langle value \rangle$ , N =  $\langle value \rangle$  and JOBVL =  $\langle value \rangle$ .  
Constraint: when JOBVL = 'V',  $LDVL \geq N$ .

IFAIL = -18

On entry, LDVR =  $\langle value \rangle$ , N =  $\langle value \rangle$  and JOBVR =  $\langle value \rangle$ .  
Constraint: when JOBVR = 'V',  $LDVR \geq N$ .

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.  
See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.  
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.  
See Section 3.6 in the Essential Introduction for further information.

## 7 Accuracy

The algorithm is backward stable for problems that are not too heavily damped, that is  $\|B\| \leq \sqrt{\|A\| \cdot \|C\|}$ .

## 8 Parallelism and Performance

F02JQF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F02JQF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

None.

## 10 Example

To solve the quadratic eigenvalue problem

$$(\lambda^2 A + \lambda B + C)x = 0$$

where

$$A = \begin{pmatrix} 2i & 4i & 4i \\ 6i & 2i & 2i \\ 6i & 4i & 2i \end{pmatrix}, \quad B = \begin{pmatrix} 3 + 3i & 2 + 2i & 1 + i \\ 2 + 2i & 1 + i & 3 + 3i \\ 1 + i & 3 + 3i & 2 + 2i \end{pmatrix} \quad \text{and} \quad C = \begin{pmatrix} 1 & 1 & 2 \\ 2 & 3 & 1 \\ 3 & 1 & 2 \end{pmatrix}.$$

The example also returns the left eigenvectors, condition numbers for the computed eigenvalues and the maximum backward errors of the computed right and left eigenpairs.

### 10.1 Program Text

```

Program f02jqfe

!      F02JQF Example Program Text
!
!      Mark 25 Release. NAG Copyright 2014.
!
!      .. Use Statements ..
!      Use nag_library, Only: f02jqf, nag_wp, x02alf
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Real (Kind=nag_wp), Parameter      :: one = 1.0E+0_nag_wp
!      Real (Kind=nag_wp), Parameter      :: tol = 0.0E0_nag_wp

```

```

Real (Kind=nag_wp), Parameter      :: zero = 0.0E+0_nag_wp
Integer, Parameter                 :: nin = 5, nout = 6
! .. Local Scalars ..
Real (Kind=nag_wp)                 :: inf, rbetaj, t0, t1, tmp
Integer                             :: i, ifail, iwarn, j, lda, ldb, ldc, &
                                     ldvl, ldvr, n, scal, sense, tdvl, tdvr

Character (1)                       :: jobvl, jobvr
! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), alpha(:), b(:,,:), beta(:), &
                                     c(:,,:), cvr(:), e(:), vl(:,,:), &
                                     vr(:,,:)

Real (Kind=nag_wp), Allocatable     :: bevl(:), bevr(:), s(:)
! .. Intrinsic Procedures ..
Intrinsic                           :: abs, aimag, cmplx, maxval, real
! .. Executable Statements ..
Write (nout,*) 'F02JQF Example Program Results'
! Skip heading in data file and read in n, scal, sense, jobVL and jobVR
Read (nin,*)
Read (nin,*) n, scal, sense
Read (nin,*) jobvl, jobvr

lda = n
ldb = n
ldc = n
ldvl = n
ldvr = n
tdvl = 2*n
tdvr = 2*n
Allocate (a(lda,n),b(ldb,n),c(ldc,n),alpha(2*n),beta(2*n),e(2*n), &
          vl(ldvl,tdvl),vr(ldvr,tdvr),s(2*n),bevr(2*n),bevl(2*n),cvr(n))

! Read in the matrices A, B and C

Read (nin,*)(a(i,1:n),i=1,n)
Read (nin,*)(b(i,1:n),i=1,n)
Read (nin,*)(c(i,1:n),i=1,n)

! Solve the quadratic eigenvalue problem
ifail = -1
Call f02jqf(scal,jobvl,jobvr,sense,tol,n,a,lda,b,ldb,c,ldc,alpha,beta, &
           vl,ldvl,vr,ldvr,s,bevl,bevr,iwarn,ifail)
If (iwarn/=0) Then
  Write (nout,*)
  Write (nout,99999) 'Warning from f02jqf. IWARN =', iwarn
End If

Write (nout,*)
If (ifail/=0) Then
  Write (nout,99999) 'Failure in f02jqf. IFAIL =', ifail
Else
! Infinity
inf = x02alf()
Do j = 1, 2*n
  rbetaj = real(beta(j))
  If (rbetaj>=one) Then
    e(j) = alpha(j)/rbetaj
  Else
    tmp = inf*rbetaj
    If ((abs(real(alpha(j)))<tmp) .And. (abs(aimag(alpha(j)))<tmp)) &
        Then
      e(j) = alpha(j)/rbetaj
    Else
      e(j) = cmplx(inf,zero,kind=nag_wp)
    End If
  End If
End Do
If (abs(e(j))<inf) Then
  Write (nout,99998) 'Eigenvalue(', j, ') = (', real(e(j)), ', ', &
                    aimag(e(j)), ', )'
Else
  Write (nout,99997) 'Eigenvalue(', j, ') is infinite'
End If

```



```

End Do

If (jobvr=='V' .Or. jobvr=='v') Then
  Write (nout,*)
  Write (nout,99996) 'Right Eigenvectors'
  Do j = 1, 2*n, 3
    If (j<=2*n-2) Then
      Write (nout,99999) 'Eigenvector ', j, '          Eigenvector ', &
        j + 1, '          Eigenvector ', j + 2
      Write (nout,99995)(vr(i,j),vr(i,j+1),vr(i,j+2),i=1,n)
    Else If (j==2*n-1) Then
      Write (nout,99999) 'Eigenvector ', j, '          Eigenvector ', &
        j + 1
      Write (nout,99994)(vr(i,j),vr(i,j+1),i=1,n)
    Else If (j==2*n) Then
      Write (nout,99999) 'Eigenvector ', j
      Write (nout,99993)(vr(i,j),i=1,n)
    End If
  End Do
End If

If (jobvl=='V' .Or. jobvl=='v') Then
  Write (nout,*)
  Write (nout,*) 'Left Eigenvectors'
  Do j = 1, 2*n, 3
    If (j<=2*n-2) Then
      Write (nout,99999) 'Eigenvector ', j, '          Eigenvector ', &
        j + 1, '          Eigenvector ', j + 2
      Write (nout,99995)(vl(i,j),vl(i,j+1),vl(i,j+2),i=1,n)
    Else If (j==2*n-1) Then
      Write (nout,99999) 'Eigenvector ', j, '          Eigenvector ', &
        j + 1
      Write (nout,99994)(vl(i,j),vl(i,j+1),i=1,n)
    Else If (j==2*n) Then
      Write (nout,99999) 'Eigenvector ', j
      Write (nout,99993)(vl(i,j),i=1,n)
    End If
  End Do
End If

If (sense==1 .Or. sense==5 .Or. sense==6 .Or. sense==7) Then
  Write (nout,*)
  Write (nout,*) 'Eigenvalue Condition numbers'
  Do j = 1, 2*n
    Write (nout,99992) s(j)
  End Do
End If

If ((sense==3) .Or. (sense==4) .Or. (sense==6) .Or. (sense==7)) Then
  t0 = maxval(bevr)
  Write (nout,*)
  Write (nout,99996) &
    'Max backward error for eigenvalues and right eigenvectors', t0
End If

If ((sense==2) .Or. (sense==4) .Or. (sense==5) .Or. (sense==7)) Then
  t1 = maxval(bevl)
  Write (nout,*)
  Write (nout,99996) &
    'Max backward error for eigenvalues and left eigenvectors ', t1
End If
End If

99999 Format (1X,3(A,I4))
99998 Format (1X,A,I3,A,1P,E11.4,A,1P,E11.4,A)
99997 Format (1X,A,I3,A)
99996 Format (1X,A,1P,E11.4)

```

```

99995 Format (3(1X,'(',1P,E11.4,',',1P,E11.4,')'))
99994 Format (2(1X,'(',1P,E11.4,',',1P,E11.4,')'))
99993 Format (1X,'(',1P,E11.4,',',1P,E11.4,')')
99992 Format (1X,1P,E11.4)
      End Program f02jqfe

```

## 10.2 Program Data

F02JQF Example Program Data

```

3      1      7      : n, scal, sense
'v'   'v'           : jobVL, jobVR

```

```

(0.0, 2.0) (0.0, 4.0) (0.0, 4.0)
(0.0, 6.0) (0.0, 2.0) (0.0, 2.0)
(0.0, 6.0) (0.0, 4.0) (0.0, 2.0) : A

```

```

(3.0, 3.0) (2.0, 2.0) (1.0, 1.0)
(2.0, 2.0) (1.0, 1.0) (3.0, 3.0)
(1.0, 1.0) (3.0, 3.0) (2.0, 2.0) : B

```

```

(1.0, 0.0) (1.0, 0.0) (2.0, 0.0)
(2.0, 0.0) (3.0, 0.0) (1.0, 0.0)
(3.0, 0.0) (1.0, 0.0) (2.0, 0.0) : C

```

## 10.3 Program Results

F02JQF Example Program Results

```

Eigenvalue( 1) = (-1.9256E+00, 1.9256E+00)
Eigenvalue( 2) = (-6.9748E-01, -1.0532E-01)
Eigenvalue( 3) = ( 1.0532E-01, 6.9748E-01)
Eigenvalue( 4) = (-4.9622E-02, -5.7288E-01)
Eigenvalue( 5) = ( 5.7288E-01, 4.9622E-02)
Eigenvalue( 6) = ( 3.9455E-01, -3.9455E-01)

```

Right Eigenvectors

```

Eigenvector 1      Eigenvector 2      Eigenvector 3
(-2.1083E-01, 7.8310E-17) ( 3.7508E-01, 1.8772E-01) ( 3.7508E-01,-1.8772E-01)
( 7.6950E-01, 0.0000E+00) ( 5.0200E-01, 2.4329E-01) ( 5.0200E-01,-2.4329E-01)
(-6.0285E-01,-4.5043E-16) ( 7.1616E-01, 0.0000E+00) ( 7.1616E-01, 0.0000E+00)
Eigenvector 4      Eigenvector 5      Eigenvector 6
(-6.5928E-01,-4.2432E-02) (-6.5928E-01, 4.2432E-02) (-3.4779E-01, 3.7418E-16)
( 3.0158E-02,-1.9723E-02) ( 3.0158E-02, 1.9723E-02) ( 8.2767E-01, 0.0000E+00)
( 7.4984E-01, 0.0000E+00) ( 7.4984E-01, 0.0000E+00) (-4.4046E-01,-7.2584E-16)

```

Left Eigenvectors

```

Eigenvector 1      Eigenvector 2      Eigenvector 3
( 1.0520E-01,-1.2569E-16) ( 7.8162E-01, 0.0000E+00) ( 7.8162E-01, 0.0000E+00)
( 7.3813E-01, 0.0000E+00) ( 5.0745E-01, 1.3518E-01) ( 5.0745E-01,-1.3518E-01)
(-6.6640E-01, 2.5141E-17) ( 3.2017E-01, 1.0381E-01) ( 3.2017E-01,-1.0381E-01)
Eigenvector 4      Eigenvector 5      Eigenvector 6
( 8.0788E-01, 0.0000E+00) ( 8.0788E-01, 0.0000E+00) ( 3.5830E-02, 1.3019E-16)
(-1.1236E-01, 3.1416E-02) (-1.1236E-01,-3.1416E-02) ( 7.0720E-01, 0.0000E+00)
(-5.7041E-01,-9.1343E-02) (-5.7041E-01, 9.1343E-02) (-7.0611E-01,-2.5620E-16)

```

Eigenvalue Condition numbers

```

3.0717E+00
6.6202E-01
6.6202E-01
2.3848E+00
2.3848E+00
1.7625E+00

```

Max backward error for eigenvalues and right eigenvectors 5.4500E-16

Max backward error for eigenvalues and left eigenvectors 5.4600E-16